
1
Introduction to Basic

Caché Basic is an easy-to-learn, object-oriented scripting language designed for rapid development
of powerful, web-based database applications. Developers familiar with Visual Basic will find it easy
to develop applications using Caché Basic.

Caché Basic is seamlessly integrated with Caché Objects and supports object, relational, and multidi-
mensional access to data.

Caché Basic is completely compatible and interoperable with Caché's other scripting language, Caché
ObjectScript. Like Caché ObjectScript code, Basic code is compiled into object code, which is executed
by a highly optimized virtual machine.

The definition of the Caché Basic is based on VBScript, enhanced by language elements necessary to
communicate with the database layer of Caché. The following table summarizes some of the key dif-
ferences between VBScript and Caché Basic:

Comparison of VBScript and Caché Basic

Caché BasicVBScriptFeature

YesYesBasic Syntax

YesYesObject Support

YesNoMultidimensional Arrays

YesNoAdditional String Operations

NoYesUser Interface Components

YesNoPortability

The main thing to keep in mind when comparing VBScript and Caché Basic is that they are designed
for different tasks: VBScript is user interface automation language designed for use within client-based

applications, such as browsers. Caché Basic is a server-based language designed to efficiently run
portable business logic for web-based, database applications.

1.1 Learning More
To find out more about Caché Basic, you can look at the following:

• The following chapters in this book provide an overview of Caché Basic.

• The Caché Basic FAQ provides a list of frequently asked questions about Caché Basic.

• The Caché Basic Reference provides descriptions of all of the operators and functions supported
by Caché Basic.

• Using Caché Objects provides a description on how the Caché Object system works.

2
Lexical Structure

Lexical structure is the set of fundamental rules that govern the behavior of a programming language.
This chapter describes the lexical structure of Caché Basic.

a.foo = 10
A.foo = 20
PrintLn "a = " & a.foo
PrintLn "A = " & A.foo

2.1 Case Sensitivity
Statement, function and operator names are case-insensitive in Basic:

PrintLn "Hello"
PRINTLN "Hello"
println "Hello" 

Variable names are case-sensitive:

a = 10
A = 20
PrintLn "a = " & a
PrintLn "A = " & A

Object identifiers (class, method, and property names) are case-insensitive, but must be locally unique
as if they were case-sensitive (in other words, you cannot have a Person and a PERSON class):

person = OpenId Sample.Person(1)
PrintLn person.Name
' PrintLn person.name  ' Error

2.2 Whitespace
Basic ignores spaces, tabs, and newlines, that appear between tokens. A token is either a function
name, variable name, number, or a keyword. Whitespace cannot appear within a token.

2.3 Comments
Comments in Basic are line oriented and start with either the Rem statement or the apostrophe ('). The
comment is always until the end of the line and everything within the comment is ignored:

' This is a comment

2.4 Literals
Literals appear within the program and represent a data value. A literal can be a constant (vbMonday).

The following example prints a variety of literals :

PrintLn "This is a String" 'String literal
PrintLn 1234 ' Integer literal
PrintLn 123.4 ' Numeric literal
PrintLn true
PrintLn vbMonday

2.5 Identifiers and Variables
An identifier is a name and can be used to name variables, functions, classes, and so on.

A variable is a named storage location that can contain data that can be modified during program
execution. Each variable has a name that uniquely identifies it within its level of scope. There are three
types of variables: local variables, process-private global variables, and global variables.

• Local variable names must begin with an alphabetic or percent character (%) followed by
alphanumeric characters. The scope of local variables is limited to the current process. Local
variables are accessible from any namespace.

• Process-private globals in Caché Basic can take either of the following syntactical forms: ̂ ||name
or ̂ |"^"|name. Other process-private global syntactical forms are not supported in Caché Basic.
The first character of a process-private global name can be either an alphabetic or percent character

(%), followed by alphanumeric characters. A process-private global name can contain period
characters; a period may not be the first or last character of the name. The scope of process-private
globals is limited to the current process. Process-private globals are accessible from any namespace.

• Globals in Caché Basic can take any of the following syntactical forms: ^name, ^|""|name, or
^|"namespace"|name. Global syntactical forms using square brackets are not supported in
Caché Basic. The first character of a global name can be either an alphabetic or percent character
(%), followed by alphanumeric characters. A global name can contain period characters; a period
may not be the first or last character of the name. The scope of globals is not limited to the process
that created it. A global is stored in the database and is therefore visible to all processes. A global
persists after the process that created it terminates. A global is specific to a namespace, and can
only be access from that namespace, or by referencing that namespace.

Variable names are case sensitive. Variable names must be unique within their scope. Variable names
may be of any length, but must be unique within the first 31 characters (exclusive of prefix characters).
Variables can have multiple levels of subscripts.

For further details on variable naming conventions, refer to the Variables chapter of Using Caché
ObjectScript. For further details on using globals as subscript arrays, see Using Multidimensional
Storage.

2.6 Labels
Within Caché Basic code you can define a label as follows:

labelOne:
    ' some code
labelTwo:
    ' some more code

A label is a valid identifier followed by a colon (:). Label names are case-sensitive. A label name can
begin with either a letter or a percent (%) character; the other characters in the label name must be
letters or numbers. A label can be of any length, but only the first 31 characters are meaningful. A
label can have the same name as a variable or a reserved word, though this is not recommended.

A label must be the first item on a program line. It can be followed by one or more commands on the
same line. A label does not have to begin at column 1, although this is generally recommended.

Labels are used by the Goto statement and the On Error Goto statement. Refer to the Error Handling
chapter for information on the On Error Goto statement.

Using Caché Basic                                                                                                                                5

Labels



2.7 Reserved Words
Within Caché Basic there are a number of reserved words that cannot be used as local variable names
or function names.

You can find the list of reserved words in the Caché Basic Reference.

3
Data Types and Values

This chapter gives a brief overview of the various data types in Basic.

3.1 Numbers
Numbers are a sequence of digits which may contain a decimal point. Basic also supports exponential
notation to represent numbers.

Some examples of numbers:

PrintLn 1
PrintLn 1.1
PrintLn 1E10

3.2 Strings
Strings in Basic are surrounded by double quotes (") and represent a sequence of letters, digits and
punctuation. The minimum length is zero and the maximum length is 32K characters.

Some examples of strings:

PrintLn "Hello"
PrintLn ""
PrintLn "Goodbye"

You can concatenate strings using the concatenation operator  “&” :

Using Caché Basic                                                                                                                                7



PrintLn "Hello" & " " & "Goodbye"

Caché Basic includes a number of specialized string operations. These are listed in the Basic Reference.

3.3 Persistent Multidimensional Arrays (Globals)
A Global is a sparse multidimensional database array. Data can be stored in a global with any number
of subscripts. Subscripts in Caché are typeless.

A global is not different from any other type of array, with the exception that the global variable name
starts with a caret (^).

Some examples of globals. First, run this example to set the global ^x:

^x = 10
PrintLn "^x has been defined"

Now, run this example to examine the value of ^x:

PrintLn "The value of ^x is: " & ^x

Similarly, since globals are arrays, you can set the value of an array reference:

^y("string",1) = 2
^y("string","a") = "b"
PrintLn "Values in ^y are " & ^y("string",1) & " and " & ^y("string","a")

For more information on arrays, see the  “Arrays”  chapter in this document. For more information on
globals, see the Using Caché Multidimensional Storage document.

3.4 Null
The Null keyword is not supported in this version of Basic.

3.5 Undefined
Variables in Caché Basic do not need to be declared or defined. By simply assigning a value to a
variable it is defined, but until this first assignment all references to this variable are undefined. You
can use the Exists function to determine if a variable is defined or undefined:

PrintLn "Does x exist? " & Exists(x)
x = 10

PrintLn "How about now? " & Exists(x)

3.6 Dates
A date value represents a valid date within the range 31 December, 1840 through 31 December, 9999.

You can use the Now function to get the current date and time:

today = Now()
PrintLn "Today is: " & today

Similarly, you can use the Date and Time functions to get the either the current date or time, respectively:

PrintLn "Date is: " & Date()
PrintLn "Time is: " & Time()

You can use the IsDate function to determine if an expression is a valid date or can be converted into
one:

today = Now()
PrintLn "IsDate: " & IsDate(today)

There are a number of functions for extracting component information from date values. These include
Day, Hour, Minute, Month, Second, Weekday, and Year:

today = Now()
PrintLn "Today is: " & today
PrintLn "Year: " & Year(today)
PrintLn "Month: " & Month(today)
PrintLn "Day: " & Day(today)
PrintLn "Weekday: " & Weekday(today)
PrintLn "Hour: " & Hour(today)
PrintLn "Minute: " & Minute(today)
PrintLn "Second: " & Second(today)

3.7 Objects
An object value refers to a instance of an in-memory object. You can assign an object value to any
local variable:

person = New Sample.Person()
PrintLn person

You can refer to the methods and properties of an object using dot syntax:

Using Caché Basic                                                                                                                                9

Dates



person.Name = "El Vez"

You can determine if a variable contains an object using the IsObject function:

str = "A string"
person = New Sample.Person()

PrintLn "Is string an object? " & IsObject(str)
PrintLn "Is person an object? " & IsObject(person)

You cannot assign an object value to a global. Doing so will result in a runtime error.

Assigning an object value to a variable (or object property) has the side effect of incrementing the
object's internal reference count. When the number of references to an object reaches 0, Caché will
automatically destroy the object (invoke its %OnClose method and remove it from memory).

4
Variables

A variable is the name of a location in which a runtime value can be stored.

4.1 Variable Typing
Variables in Caché Basic are, like in Caché ObjectScript, JavaScript and VBScript, untyped. This
means that you can assign a string value to a variable and, later on, a numeric value to the same variable.
Caché automatically converts the value of variables to other types, based on the context in which it is
used.

Conversion amongst the various types are carried out using the same rules as Caché ObjectScript.

Caché Basic Type Conversion Rules

RulesToFrom

Not allowed.ObjectNumber

The number is converted to a string of
numeric characters that corresponds to the
number's value: 22 becomes “22” .

StringNumber

The object reference is converted to an
integer whose value is the object instance
number.

NumberObject

The object reference is converted to a
string in the form: “oref@classname”
where oref is the object instance number
and classname is the type of object.

StringObject

The string is parsed left-to-right until a
non-numeric character is encountered.The
leading numeric characters are converted
to the corresponding numeric value. If
there are no leading numeric characters,
the string is converted to 0. For example,
“123”  converts to 123, “123abc”  converts
to 123, and “abc123”  converts to 0.

NumberString

Not allowed.ObjectString

4.2 Variable Declaration
Declaring variables in Caché Basic is not required. If you want to take advantage of detecting undeclared
variables by the compiler you can use the Option Explicit clause at the beginning of your program.
In that case you must declare all variables using the Dim statement:

Option Explicit
Dim a

4.3 Variable Scope
Variables used in Caché Basic are visible within the routine, unless a procedure or function explicitly
declares the variable using the Dim statement.

5
Operators and Expressions

This chapter provides an overview of expressions and operators within Caché Basic.

5.1 Expressions
An expression is a combination of keywords, operators, variables, and constants that yield a string,
number, or object value. An expression can perform a calculation, manipulate characters, or test data.

5.2 Operators
Caché Basic includes a variety of built-in operators.

5.2.1 Operator Precedence

When several operations occur in an expression, each part is evaluated and resolved in a predetermined
order called operator precedence. Parentheses can be used to override the order of precedence and
force some parts of an expression to be evaluated before other parts. Operations within parentheses
are always performed before those outside. Within parentheses, however, normal operator precedence
is maintained.

When expressions contain operators from more than one category, arithmetic operators are evaluated
first, comparison operators are evaluated next, and logical operators are evaluated last. Comparison
operators all have equal precedence; that is, they are evaluated in the left-to-right order in which they
appear. Arithmetic and logical operators are evaluated in the following order of precedence:

LogicalComparisonArithmetic

NotEquality (=)Exponentiation (^)

AndInequality (<>)Negation (-)

OrLess than (<)Multiplication and division (*, /)

XorGreater than (>)Integer division (\)

EqvLess than or equal to (<=)Modulus arithmetic (Mod)

ImpGreater than or equal to (>=)Addition and subtraction (+, -)

&IsString concatenation (&)

When multiplication and division occur together in an expression, each operation is evaluated as it
occurs from left to right. Likewise, when addition and subtraction occur together in an expression,
each operation is evaluated in order of appearance from left to right.

The string concatenation operator (&) is not an arithmetic operator, but in precedence it does fall after
all arithmetic operators and before all comparison operators. The Is operator is an object reference
comparison operator. It does not compare objects or their values; it checks only to determine if two
object references refer to the same object.

5.3 Comparison Operators
The Is operator has specific comparison functionality that differs from the operators in the following
table. The following table contains a list of the comparison operators and the conditions that determine
whether result is True or False:

False ifTrue ifDescriptionOperator

expression1 >=
expression2

expression1 <
expression2

Less than<

expression1 >
expression2

expression1 <=
expression2

Less than or equal
to

<=

expression1 <=
expression2

expression1 >
expression2

Greater than>

expression1 <
expression2

expression1 >=
expression2

Greater than or
equal to

>=

expression1 <>
expression2

expression1 =
expression2

Equal to=

expression1 =
expression2

expression1 <>
expression2

Not equal to<>

When comparing two expressions, you may not be able to easily determine whether the expressions
are being compared as numbers or as strings.

The following table shows how expressions are compared or what results from the comparison,
depending on the underlying subtype:

ThenIf

Perform a numeric comparison.Both expressions are numeric

Perform a string comparison.Both expressions are strings

The numeric expression is less than the string
expression.

One expression is numeric and
the other is a string

6
Statements

Programs consist of a series of statements. Simple statements assign or retrieve values to and from
variables, while complex statements like If and loops support program flow or provide powerful
functionality using objects.

6.1 Expression Statements
Expression statements are used to assign values to variables or apply an operation. For example:

msg = "Hello"
result = obj.Method()
Print x

An assignment expression sets the value of a variable, array element, or object property:

var = 1
array(1) = 2
obj.Property = 3

The end of a statement is normally marked by a carriage return. You can, however, place multiple
statements on the same line by placing the colon (:) character between them:

a = 1 : b = 2 : c = 3
Print a : Print " " : Print b : Print " " : Print c

6.2 Call
The Call statement lets you call a function or method from within Basic:

Call obj.Method(a,b)

Within a Call statement, the actual  “Call”  token is completely optional and can be omitted:

obj.Method(a,b)

By default, all function arguments are passed by reference within Basic. You can override this default
on a argument-by-argument basis using the ByRef and ByVal directives:

obj.Method(ByRef a, ByVal b)

6.3 If
The If statement provides the ability to execute statements conditionally. Within Caché Basic, an If
statement takes the following form:

If (expression) Then
    statement()
End If

If expression evaluates to true, the body of the If statement is executed. The body of the If statement
consists of a block of one or more statements between a Then and End If tokens.

There is also a simpler, single-line form of the If statement:

If (expression) Then statement()

In the single-line form, there is no End If token; the body of the If statement follows the If token and
continues to the end of the line. Note that you can include multiple statements by separating them with
colon (:) characters:

If (expression) Then statement1() : statement2()

An If statement make contain an Else clause:

If (expression) Then
    statement1()
Else
    statement2()
End If

The statements within the Else block are executed if expression evaluates to False.

6.4 ElseIf
ElseIf is an optional statement which is part of the closest If statement. The ElseIf statement allows
an If statement to sequentially test a set of expressions:

If (expression) Then
    statement1()
ElseIf (expression2) Then
    statement2()
ElseIf (expression3) Then
    statement3()
Else
    statement3()
End If

The statements within the first If or ElseIf block whose expression is true are executed. If none are
true, then the Else block is executed.

6.5 For
The For statement is used to repeat a sequence of statements.

6.5.1 For...Next

The For...Next loop uses a counter to execute a block of statements a specified number of times:

For i = 1 To 10
    PrintLn i
Next

In this case, i will be initialized to 1 and the body of the For...Next loop will be executed until i reached
10. The value of i is incremented by 1 after each iteration.

Using the Step keyword, you can increase or decrease the counter variable by the value you specify.
To decrease the counter variable, use a negative Step value. You must specify an end value that is less
than the start value:

For i = 10 To 1 Step -1
    PrintLn i
Next
PrintLn "Blast Off!"

If you like (say for legibility), you can place the counter variable after the Next token:

For i = 1 To 10
    PrintLn i
Next i

Using the wrong variable after the Next token results in a syntax error.

6.5.2 For Each...Next

The For Each...Next repeats a group of statements for each item in a collection or each element of an
array. A For Each...Next loop is similar to a For...Next loop. Instead of repeating the statements a
specified number of times, a For Each...Next loop repeats a group of statements for each item in a
collection of objects or for each element of an array. This is especially helpful if you don't know how
many elements are in a collection.

For example, the following creates an array of ten items and then uses a For Each...Next loop to display
the subscript of each item:

' Create a simple array
For i = 1 To 10
    array(i) = i * i
Next

' Now display the subscripts and contents of the array
For Each key in array
    PrintLn key & ": " & array(key)
Next

With a multidimensional array, this gets more interesting:

' Create a multidimensional array
For i = 1 To 5
    For j = 1 To 5
        array(i, j) = i * j
    Next j
Next i

' Now display the subscripts and contents of the array
For Each key1 in array
    For Each key2 in array(key1)
        PrintLn key1 & "," & key2 & ": " & array(key1,key2)
    Next key2
Next key1

6.6 While
You can use a While statement to run a block of statements an indefinite number of times:

While (expression)
    statement()
Wend

The statements are repeated as long as expression is True. The end of the While block is marked by
the Wend token.

For example, the following code executes its While block 9 times:

x = 1
While (x < 10)
    PrintLn x
    x = x + 1
Wend

6.7 Do...While
You can use a Do...While statement to run a block of statements an indefinite number of times:

Do
    statement()
Loop While (expression)

The statements are repeated as long as expression is True. The Do...While block is contained within
the Do and Loop tokens.

The Do...While statement differs from the While statement in that it executes its contents before
evaluating its loop expression.

6.8 Select ... Case
The Select...Case statement lets a program choose one of several execution paths based on evaluating
an expression. It is a convenient alternative to a long series of ElseIf statements.

A Select...Case statement takes the following form:

Select Case testexpression
    [Case expressionlist-n
        [statements-n]] . . .
    [Case Else expressionlist-n
        [elsestatements-n]]
End Select

The expression testexpression is evaluated and its result is then compared with each of the expressionlist
values associated with each Case clause. The code block associated with the first matching Case clause
is then executed. If there is no match, then the Case Else block (if present) is executed.

For example:

x = 2
Select Case x
    Case 1
        PrintLn "First case"
    Case 2
        PrintLn "Second case"
    Case Else
        PrintLn "Unknown"
End Select

There is also a simpler form of Case that can be used as an in-line expression:

Case(expr, val1:ret1, val2:ret2)

In this case (pun intended), the expression expr is first evaluated, and then its value is compared with
each of a series of values (val1, val2, in this example). The value associated with the first matching
value is returned.

For example:

x = 2
PrintLn Case(x,1:"hello",2:"goodbye")

6.9 Dim
To declare local variables Caché Basic supports the Dim statement. There is no need to declare variables
in Caché Basic and the declaration is optional unless you include the Option Explicit clause in your
program.

6.10 Return
You can return a value from a method using the Return statement:

Method Add(a As %Integer, b As %Integer) As %Integer [language = basic]
{
    Return a + b
}

Note that it is a syntax error to use the Return statement within a method that does not return a value.

7
User-defined Basic Code

A typical application consists of some number of user-defined methods and functions which contain
the actual application logic.

7.1 Basic Methods
Within a Caché Basic application, the user-defined business logic (the code for the application) is
typically contained within methods defined within classes:

Class MyApp.MyClass [language = basic]
{

Method MyMethod() As %Integer
{
    a = 22
    Return a
}

}

Note that the primary language for this class is specified (using the language keyword) as  “Basic” .
This means that all methods will be implemented using Basic. You can override this on a method-by-
method basis and it is possible to have mix methods using different languages together within the same
class.

You can call a Basic method from Basic:

x = obj.MyMethod()

or from ObjectScript:

 Set x = obj.MyMethod()

7.2 Basic Routines and Functions
You can create a Basic routine (similar to an ObjectScript routine but implemented using Basic) and
define one or more functions within it:

 'MyRoutine.BAS

Function Add(a,b)
    Return a + b
End Function

You can call this function from Basic (from outside the MyRoutine.BAS routine):

x = Add@MyRoutine(2,3)

or from ObjectScript:

 Set x = $$Add^MyRoutine(2,3)

7.3 Calling Caché ObjectScript Functions
Should you need to, it is possible to call Caché ObjectScript functions from Basic.

Note that you can call object methods implemented in ObjectScript from Basic and vice versa using
normal object syntax.

7.3.1 ObjectScript Extrinsic Functions

Should you need to call a Caché ObjectScript user-defined (extrinsic) function from Basic, you can
do this using a special syntax:

val = Func@Routine(args)

Where Func is the name of the function and Routine is the name of the routine containing the function.

For example, suppose you have a Caché ObjectScript routine, Math.INT:

  ; Math.INT
Add(a,b) PUBLIC {
    Quit a + b
}

You can call this function from Caché Basic as follows:

val = Add@Math(a,b)

This is equivalent to the following Caché ObjectScript code:

 Set val = $$Add^Math(.a,.b)

Note the following differences between Basic and ObjectScript:

1. There is no Set command needed in Basic.

2. You do not need to use $$ in Basic for functions that return values.

3. Basic uses the  “@”  character as a delimiter between the function and routine names.

4. By default, all function arguments are passed by reference in Basic.

If you want to pass arguments by value in Basic, you need to use the ByVal directive:

val = Add@Math(ByVal a, ByVal b)

7.3.2 ObjectScript Intrinsic Functions

You cannot directly call built-in (intrinsic) Caché ObjectScript functions (such as the various $zutil
functions) from Basic. You can, however, create a simple Caché ObjectScript function or method that
invokes the desired function and call that from Basic.

For example, the following ObjectScript function invokes $ZUTIL(5) to get the current namespace:

  ; MyUtils.INT
GetNamespace() PUBLIC {
 ; Return the current namespace name
    Quit $ZUTIL(5)
}

You can call this from Caché Basic as follows:

PrintLn "Current Namespace: " & GetNamespace@MyUtils()

Using Caché Basic                                                                                                                              27

Calling Caché ObjectScript Functions





8
Arrays

Caché Basic includes much more powerful support for arrays than other versions of Basic. Specifically,
with Caché Basic you can:

• Create sparse, multidimensional arrays.

• Use any combination of integer, string, and numeric values are array subscripts. Such arrays are
automatically sorted by subscript value; you never need to invoke a sort function.

• Use arrays without having to declare their size and dimensions. You do not need to Dim or Redim
arrays in Caché Basic.

• Use a number of built-in commands for manipulating and traversing arrays.

In addition, you can create and work with persistent multidimensional arrays (globals) that are auto-
matically stored within the database. For more information, refer to Using Caché Multidimensional
Storage.

8.1 Defining Arrays
You do not need to declare array variables in any special way. To create an array, you simply place
values into it:

' Create an array
arr(1) = "First element"
arr(2) = "Second element"
arr(3) = "Third element"

' Display the contents of the array
For i = 1 To 3
    PrintLn i & ": " & arr(i)
Next

Using Caché Basic                                                                                                                              29



Each array element (node) can contain up to 32K of data.

To create a multidimensional array, simply use more subscripts:

' Create a multidimensional array
arr(1,1) = "X" : arr(1,2) = "O" : arr(1,3) = "X"
arr(2,1) = "X" : arr(2,2) = "X" : arr(2,3) = "O"
arr(3,1) = "O" : arr(3,2) = "O" : arr(3,3) = "X"

' Display the contents of the array
For row = 1 To 3
    For col = 1 To 3
        Print arr(row,col) & " "
    Next col
    PrintLn
Next row

You can use strings as array subscripts:

' Create an associative array
notes("Jack") = "Fell down; Broke crown"
notes("Jill") = "Tumbled after"

' Display the contents of the array
PrintLn "Jack: " & notes("Jack")
PrintLn "Jill: " & notes("Jill")

8.1.1 Existence Testing

You can test if an array node (or any variable) is defined (has a value) or has sub-nodes using the
Exists function. The Exists function returns one of the following possible values:

Exists Return Values

MeaningValueConstant

Variable is undefined.0vbUndefined

Variable has a value.1vbHasValue

Variable is not defined but has
subnodes.

2vbHasArray

Variable is defined and has
subnodes.

3vbHasValue AND
vbHasArray

For example:

' Initialize
var = "Variable"
arr(1) = "Array node with subnode"
arr(1,1) = "Array node"

' Test for existence
PrintLn "Exists(a)        " & Exists(a)
PrintLn "Exists(var)      " & Exists(var)
PrintLn "Exists(arr)      " & Exists(arr)
PrintLn "Exists(arr(1))   " & Exists(arr(1))
PrintLn "Exists(arr(1,1)) " & Exists(arr(1,1))

8.1.2 Deleting Arrays

You can delete an array (its subscripts and values) using the Erase command:

array = "root node"
array("subnode") = "subnode"
array("subnode", "subnode") = "subnode, subnode"
Println Exists(array) 'returns 3; variable defined and has array elements
Erase array 
Println Exists(array) 'returns 0

8.2 Traversing Arrays
There are a number of ways to traverse an array to find what subscripts and values it contains.

8.2.1 Traversing with a For...Next Loop

If an array is subscripted by integer values and you know how many elements it contains, you can use
a simple For...Next loop to list its contents:

' Create an array
For i = 1 To 10
    arr(i) = "Item " & i
Next i

' Display the contents of the array
For i = 1 To 10
    PrintLn i & ": " & arr(i)
Next

8.2.2 Traversing with a For Each...Next Loop

You can use a For Each...Next loop to list all the subscripts for an array:

' Create an array
food("Apple") = "Fruit"
food("Banana") = "Fruit"
food("Asparagus") = "Vegetable"
food("Broccoli") = "Vegetable"

' Display the array subscripts
For Each key In food
    PrintLn key
Next

8.2.3 The Traverse Function

You can use the Traverse function to find the next (or previous) subscript at any point within an array.
(Traverse is the Basic equivalent of the Caché ObjectScript $Order function).

The Traverse function takes an array reference and returns the next (or previous) subscript at the
specified subscript level. The specified subscript level is simply the last subscript listed in the array
reference. For example:

' Define an array
arr("a",1) = "a1"
arr("a",2) = "a2"
arr("b",1) = "b1"
arr("b",2) = "b2"

PrintLn Traverse(arr(""))
PrintLn Traverse(arr("a"))
PrintLn Traverse(arr("b"))

PrintLn Traverse(arr("a",""))
PrintLn Traverse(arr("a",1))

We can use Traverse in a While loop to find all of the subscripts of the food array from the previous
section:

' Create an array
food("Apple") = "Fruit"
food("Banana") = "Fruit"
food("Asparagus") = "Vegetable"
food("Broccoli") = "Vegetable"

' Display the array subscripts
key = Traverse(food("")) ' find first subscript
While (key <> "")
    PrintLn key
    key = Traverse(food(key)) ' find next subscript
Wend

To find previous nodes with Traverse, use the optional direction flag as a second argument:

' Create an array
food("Apple") = "Fruit"
food("Banana") = "Fruit"
food("Asparagus") = "Vegetable"
food("Broccoli") = "Vegetable"

' Display the array subscripts backwards
key = Traverse(food(""),-1) ' find last subscript
While (key <> "")
    PrintLn key
    key = Traverse(food(key),-1) ' find previous subscript
Wend

9
Using Objects

Caché includes a full-featured, next-generation object database specifically designed to meet the needs
of complex, transaction oriented applications.

The Caché Object Model includes the following features:

• Classes—You can define classes that represent the state (data) and behavior (code) of your
application components. Classes are used to create instances of objects as both runtime components
and as items stored within the database.

• Properties—Classes can include properties, which specify the data associated with each object
instance. Properties can be simple literals (such as strings or integers), user-defined types (defined
using data type classes), complex (or embedded) objects, collections, or references to other objects.

• Relationships—Classes can define how instances of objects are related to one another. The system
automatically provides navigational methods for relationships as well as referential integrity within
the database.

• Methods—Classes can define behavior by means of methods: executable code associated with an
object. Object methods run within a Caché server process (though they can be invoked from a
remote client). Object methods can be scripted using Caché ObjectScript, SQL, or they can be
generated using method generators, which are code that automatically creates customized methods
according to user-defined rules.

• Inheritance—By deriving new classes from existing ones, you can reuse previously written code
as well as create specialized versions of classes.

• Polymorphism—Caché supports complete object polymorphism. This means that applications
can use a well-defined interface (a set of methods and properties provided by a superclass) and
the system will automatically invoke the correct interface implementation based on the type of
each object. This makes it much easier to develop flexible database applications.

� Swizzling (also known as �lazy loading�)�Caché automatically swizzles (brings into memory
from disk) any related persistent objects when they are referenced from other objects. This greatly
simpli�es working with complex data models.

For detailed information on how to use Objects in Caché please refer to Using Caché Objects.

9.1 Defining Classes

An object is an instance of a speci�c class. Within Caché you can de�ne classes in several ways. For
information on how to de�ne classes refer to one of the following:

� Caché Classes in Using Caché Objects.

� Using Caché Studio.

9.2 Creating Object Instances

You can create an object instance by either creating a new instance using the New command or you
can open a persistent object previously stored in the database using the OpenId command.

9.2.1 The New Command

The New command creates and returns a new instance of the speci�ed class. For example:

person = New Sample.Person()
person.Name = "El Vez"
PrintLn person.Name

The name of the class (with an optional package name) follows the New command. There are paren-
theses following the class name. If you place an argument within these parentheses, then it will be
passed to the constructor (%OnNew method) for the object.

The New command has exactly the same behavior as invoking the %New method in Caché ObjectScript.

9.2.2 The OpenId Command

The OpenId command �nds, opens, and returns an instance of a persistent class that is stored within
the database using its unique object identi�er value. For example:

person = OpenId Sample.Person(22)
PrintLn "Name: " & person.Name
PrintLn "Age: " & person.Age

